'Set fuse bit for 1 Mhz clock

'Setup Routine **************

'Declare variables

 Dim Batteryvoltage As Word                                 'Measured battery voltage

 Dim Timera As Word                                         'Timer for Constant Voltage charge phase

 Dim Adclowbyte As Byte                                     'A-D converter low byte

 Dim Adchighbyte As Byte                                    'A-D converter High byte

 Dim Portcpinsdirection As Byte                             'To Setup Port C

 Dim N As Integer                                           'Used in "for next" loops

 Dim C As Bit                                                'To end constant current phase

 Dim DisplayCounter as Byte                                  'To control Charge LED display

'Configure I/O ports

 Config Portb.0 = Input                                    'Sense 6A current indication

 Config Portb.1 = Output                                   'Indicate up/down for PWM Processor

 Config Portb.2 = Input                                    'Sense 7A over current indication

 Config Portb.3 = Output                                   'Mains and Battery relays control

 Config Portd = Output                                     'Charge LEDs

 Portcpinsdirection = &B1111                                 'PortC Pins 0 to 3 are used as battery Charge Status indicators.

 Out &H34 , Portcpinsdirection

'Clear variables

 C=0

 DisplayCounter = 0                                         'Clear Constant Current Phase indicator

 Timera = 0                                                 'ClearTimer

 Batteryvoltage = 0                                         'Clear Batteryvoltage

 Portb.1 = 0                                                'PWM off

 Portb.3 = 0                                                'Mains and Battery relays off

 Portd = &B00000000                                         'Charge LEDs off

 Out &H35 , &B0000                                          'Port c=0 all battery status indicators off

 wait 1                                                     'Wait for PWM processor to reduce PWM to zero

'Check for battery fault

  Portb.3 = 1                                               'Switch Mains and Battery relays on

  wait 1

   For N = 1 To 100

   Out &H27 , &B00000101                                    'Setup the Adc - Port ADC5

   Out &H26 , &B11010111

   Adclowbyte = Inp(&H24)                                   'Read Battery Voltage

   Adchighbyte = Inp(&H25)

   Batteryvoltage = Makeint(adclowbyte , Adchighbyte)       'Combine ADC bytes into a word for evaluation

   Next N

 If Batteryvoltage < &B110000011 Then Goto Fault           'Low battery fault display (11V)

 If Batteryvoltage > &B1000010001 Then Goto Fault           'High battery fault display (15V)

 If Pinb.2=1 then goto Fault                                'Current Processor will indicate "Over 7A".

'Main Routine **************

Start:

   Portb.3 = 1

                                                            'Put Mains and Battery relays on

   For N = 1 To 100

   Out &H27 , &B00000101                                    'Setup the Adc - Port ADC5

   Out &H26 , &B11010111

   Adclowbyte = Inp(&H24)                                   'Read Battery Voltage

   Adchighbyte = Inp(&H25)

   Batteryvoltage = Makeint(adclowbyte , Adchighbyte)       'Combine ADC bytes into a word for evaluation

   Next N

   If Batteryvoltage < &B110000011 Then Gosub Fault        'Low battery fault display (11V)

   If Batteryvoltage > &B1000010001 Then Goto Fault         'High battery fault display (15V)

   If Pinb.2=1 then goto Fault                              'Current Processor will indicate "Over 7A".

   Gosub ChargeDisplay

   If C = 1 then gosub ConstantVoltage                      'Constant Voltage Phase

   If C = 1 then Goto Start

   If Batteryvoltage <= &B111111000 Then Gosub ContantCurrent       'Regulate current on 6A

   If Batteryvoltage > &B111111000 Then Gosub ConstantVoltage       'Regulate voltage on 14.3V

 Goto Start

'Sub-Routines *************

ContantCurrent:

  If pinb.0=0 then portb.1=1                                'Current Processor indicates "Under 6A" PWM on

  If pinb.0=1 then portb.1=0                                'Current Processor indicates "Over 6A" PWM off

  waitms 50

  Portd = &B00000010                                         'Put Constant Current Phase LED on

  Return

ConstantVoltage:

  C=1

  If Batteryvoltage <= &B111111000 Then portb.1=1         'Battery Voltage under 14.3V, PWM on

  If Batteryvoltage > &B111111000 Then portb.1=0          'Battery Voltage over 14.3V, PWM off

  Timera = Timera + 1                                       'Increment Constant Voltage Phase Timer

  If Timera > 26000 then goto Chargend                       'Upon time complete, end charging (3 hours)

  waitms 50

  Portd = &B00000100                                        'Put Constant Voltage Phase LED on

  Return

Fault:

 Out &H35 , &B0000                                          'Reset Display Counter

 Portd = &B00000001                                         'Put Fault LED on

 Portb.1 = 0                                                'PWM Off

 Portb.3 = 0                                                'Put Mains and Battery relays off

 Waitms 50

 Portd = &B00000000                                         'Put Fault LED off

 Waitms 50

Goto Fault

ChargeDisplay:

                                                            'Used to indicate charging process is in progress

  DisplayCounter =  DisplayCounter + 1                      'Increase Display Counter

  If  DisplayCounter > 4 then Out &H35 , &B0000

  If  DisplayCounter > 4 then DisplayCounter = 0           'Reset Display Counter

  waitms 50

  If DisplayCounter = 1 then Out &H35 , &B0001             'Control Charge Indicator

  If DisplayCounter = 2 then Out &H35 , &B0010

  If DisplayCounter = 3 then Out &H35 , &B0100

  If DisplayCounter = 4 then Out &H35 , &B1000

 Return

Chargend:

 Portb.1 = 0                                                'Switch PWM off

 Portb.3 = 0                                                'Switch Mains and Battery relays off

 Portd = &B00001000                                         'Put Charge Complete LED on

 Out &H35 , &B0000                                          'Reset Display Counter

 End  

